

PROFESSIONAL **CMAKE**

A PRACTICAL GUIDE



CRAIG SCOTT

Professional CMake: A Practical Guide

14th Edition

ISBN 978-1-925904-22-2

© 2018-2023 by Craig Scott

This book or any portion thereof may not be reproduced in any manner or form without the express written permission of the author, with the following specific exceptions:

- The original purchaser may make personal copies exclusively for their own use on their electronic devices, provided that all reasonable steps are taken to ensure that only the original purchaser has access to such copies.
- Permission is given to use any of the code samples in this work without restriction. Attribution is not required.

The advice and strategies contained within this work may not be suitable for every situation. This work is sold with the understanding that the author is not held responsible for the results accrued from the advice in this book.

<https://crascit.com>

Table of Contents

Preface	1
Acknowledgments	1
Part I: Fundamentals	3
1. Introduction	4
2. Setting Up A Project	6
2.1. In-source Builds	6
2.2. Out-of-source Builds	7
2.3. Generating Project Files	7
2.4. Running The Build Tool	8
2.5. Recommended Practices	9
3. A Minimal Project	10
3.1. Managing CMake Versions	10
3.2. The project() Command	12
3.3. Building A Basic Executable	13
3.4. Commenting	13
3.5. Recommended Practices	14
4. Building Simple Targets	15
4.1. Executables	15
4.2. Defining Libraries	16
4.3. Linking Targets	17
4.4. Linking Non-targets	19
4.5. Old-style CMake	19
4.6. Recommended Practices	20
5. Variables	22
5.1. Variable Basics	22
5.2. Environment Variables	24
5.3. Cache Variables	24
5.4. Scope Blocks	26
5.5. Potentially Surprising Behavior Of Variables	28
5.6. Manipulating Cache Variables	29
5.6.1. Setting Cache Values On The Command Line	29
5.6.2. CMake GUI Tools	30
5.7. Printing Variable Values	34
5.8. String Handling	34
5.9. Lists	37
5.9.1. Problems With Unbalanced Square Brackets	39
5.10. Math	40
5.11. Recommended Practices	41

6. Flow Control	42
6.1. The if() Command	42
6.1.1. Basic Expressions	42
6.1.2. Logic Operators	44
6.1.3. Comparison Tests	44
6.1.4. File System Tests	46
6.1.5. Existence Tests	48
6.1.6. Common Examples	49
6.2. Looping	50
6.2.1. foreach()	50
6.2.2. while()	53
6.2.3. Interrupting Loops	53
6.3. Recommended Practices	55
7. Using Subdirectories	56
7.1. add_subdirectory()	56
7.1.1. Source And Binary Directory Variables	57
7.1.2. Scope	58
7.1.3. When To Call project()	60
7.2. include()	61
7.3. Project-relative Variables	63
7.4. Ending Processing Early	65
7.5. Recommended Practices	67
8. Functions And Macros	69
8.1. The Basics	69
8.2. Argument Handling Essentials	70
8.3. Keyword Arguments	72
8.4. Returning Values	78
8.4.1. Returning Values From Functions	78
8.4.2. Returning Values From Macros	80
8.5. Overriding Commands	81
8.6. Special Variables For Functions	83
8.7. Other Ways Of Invoking CMake Code	84
8.8. Problems With Argument Handling	87
8.8.1. Parsing Arguments Robustly	89
8.8.2. Forwarding Command Arguments	90
8.8.3. Special Cases For Argument Expansion	92
8.9. Recommended Practices	95
9. Properties	97
9.1. General Property Commands	97
9.2. Global Properties	101
9.3. Directory Properties	101

9.4. Target Properties	102
9.5. Source Properties	103
9.6. Cache Variable Properties	105
9.7. Other Property Types	106
9.8. Recommended Practices	106
10. Generator Expressions	108
10.1. Simple Boolean Logic	109
10.2. Target Details	111
10.3. General Information	112
10.4. Path Expressions	113
10.5. Utility Expressions	114
10.6. Recommended Practices	115
11. Modules	117
11.1. Checking Existence And Support	118
11.2. Other Modules	125
11.3. Recommended Practices	125
12. Policies	127
12.1. Policy Control	127
12.2. Policy Scope	130
12.3. Recommended Practices	132
13. Debugging And Diagnostics	134
13.1. Log Messages	134
13.1.1. Log Levels	134
13.1.2. Message Indenting	136
13.1.3. Message Contexts	137
13.1.4. Check Messages	139
13.2. Color Diagnostics	141
13.3. Print Helpers	142
13.4. Tracing Variable Access	143
13.5. Debugging Generator Expressions	143
13.6. Profiling CMake Calls	145
13.7. Discarding Previous Results	145
13.8. Recommended Practices	146

Part II: Builds In Depth **148**

14. Build Type	149
14.1. Build Type Basics	149
14.1.1. Single Configuration Generators	150
14.1.2. Multiple Configuration Generators	150
14.2. Common Errors	151
14.3. Custom Build Types	152

14.4. Recommended Practices	156
15. Compiler And Linker Essentials	158
15.1. Target Properties	158
15.1.1. Compiler Flags	158
15.1.2. Linker Flags	160
15.1.3. Sources	162
15.2. Target Property Commands	162
15.2.1. Linking Libraries	162
15.2.2. Linker Options	163
15.2.3. Header Search Paths	164
15.2.4. Compiler Defines	165
15.2.5. Compiler Options	165
15.2.6. Source Files	165
15.3. Directory Properties And Commands	167
15.4. De-duplicating Options	170
15.5. Compiler And Linker Variables	171
15.5.1. Compiler And Linker Variables Are Single Strings, Not Lists	173
15.5.2. Distinguish Between Cache And Non-cache Variables	173
15.5.3. Prefer Appending Over Replacing Flags	174
15.5.4. Understand When Variable Values Are Used	174
15.6. Language-specific Compiler Flags	175
15.7. Compiler Option Abstractions	176
15.7.1. Warnings As Errors	176
15.7.2. System Header Search Paths	177
15.7.3. Runtime Library Selection	180
15.7.4. Debug Information Format Selection	181
15.8. Recommended Practices	183
16. Advanced Linking	186
16.1. Require Targets For Linking	186
16.2. Customize How Libraries Are Linked	187
16.2.1. Link Group Features	188
16.2.2. Link Library Features	188
16.2.3. Custom Features	190
16.2.4. Feature Validity	190
16.3. Propagating Up Direct Link Dependencies	190
16.3.1. Link Seaming Example	191
16.3.2. Static Plugins	194
16.4. Recommended Practices	194
17. Language Requirements	195
17.1. Setting The Language Standard Directly	195
17.2. Setting The Language Standard By Feature Requirements	198

17.2.1. Detection And Use Of Optional Language Features	199
17.3. Recommended Practices	200
18. Target Types	202
18.1. Executables	202
18.2. Libraries	203
18.2.1. Basic Library Types	203
18.2.2. Object Libraries	203
18.2.3. Imported Libraries	204
18.2.4. Interface Libraries	206
18.2.5. Interface Imported Libraries	208
18.2.6. Library Aliases	209
18.3. Promoting Imported Targets	211
18.4. Recommended Practices	212
19. Custom Tasks	215
19.1. Custom Targets	215
19.2. Adding Build Steps To An Existing Target	218
19.3. Commands That Generate Files	219
19.4. Configure Time Tasks	223
19.5. Platform Independent Commands	226
19.6. Combining The Different Approaches	228
19.7. Recommended Practices	230
20. Working With Files	232
20.1. Manipulating Paths	232
20.1.1. <code>cmake_path()</code>	232
20.1.2. Older Commands	235
20.2. Copying Files	238
20.3. Reading And Writing Files Directly	246
20.4. File System Manipulation	250
20.5. File Globbing	252
20.6. Downloading And Uploading	254
20.7. Recommended Practices	256
21. Specifying Version Details	258
21.1. Project Version	258
21.2. Source Code Access To Version Details	260
21.3. Source Control Commits	263
21.4. Recommended Practices	266
22. Libraries	268
22.1. Build Basics	268
22.2. Linking Static Libraries	269
22.3. Shared Library Versioning	269
22.4. Interface Compatibility	271

22.5. Symbol Visibility	276
22.5.1. Specifying Default Visibility	277
22.5.2. Specifying Individual Symbol Visibilities	277
22.6. Mixing Static And Shared Libraries	283
22.7. Recommended Practices	286
23. Toolchains And Cross Compiling	288
23.1. Toolchain Files	289
23.2. Defining The Target System	290
23.3. Tool Selection	291
23.4. System Roots	294
23.5. Compiler Checks	295
23.6. Examples	296
23.6.1. Raspberry Pi	296
23.6.2. GCC With 32-bit Target On 64-bit Host	296
23.7. Android	297
23.7.1. Historical Context	297
23.7.2. Using The NDK	298
23.7.3. Android Studio	299
23.7.4. ndk-build	300
23.7.5. Visual Studio Generators	300
23.8. Recommended Practices	300
24. Apple Features	302
24.1. CMake Generator Selection	302
24.2. Application Bundles	305
24.2.1. Bundle Structure	305
24.2.2. Bundle Info.plist Files	306
24.2.3. Sources, Resources And Other Files	309
24.3. Frameworks	310
24.3.1. Framework Structure	311
24.3.2. Framework Info.plist Files	312
24.3.3. Headers	313
24.4. Loadable Bundles	314
24.5. Build Settings	314
24.5.1. SDK Selection	314
24.5.2. Deployment Target	315
24.5.3. Device Families	316
24.5.4. Compiler Test Workarounds	317
24.6. Code Signing	318
24.6.1. Signing Identity And Development Team	318
24.6.2. Provisioning Profiles	319
24.6.3. Entitlements	320

24.7. Creating And Exporting Archives	320
24.8. Universal Binaries	322
24.8.1. Device-only Bundles	323
24.8.2. Intel And Apple Silicon Binaries	324
24.8.3. Combined Device And Simulator Binaries	325
24.9. Linking Frameworks	326
24.10. Embedding Frameworks	327
24.11. Limitations	330
24.12. Recommended Practices	331

Part III: The Bigger Picture **333**

25. Finding Things	334
25.1. Finding Files And Paths	334
25.1.1. Apple-specific Behavior	339
25.1.2. Cross-compilation Controls	339
25.1.3. Validators	341
25.2. Finding Paths	341
25.3. Finding Programs	342
25.4. Finding Libraries	343
25.5. Finding Packages	345
25.5.1. Package Registries	353
25.5.2. FindPkgConfig	355
25.6. Ignoring Search Paths	358
25.7. Debugging find_...() Calls	359
25.8. Recommended Practices	359
26. Testing	363
26.1. Defining And Executing A Simple Test	363
26.2. Test Environment	366
26.3. Pass / Fail Criteria And Other Result Types	368
26.4. Test Grouping And Selection	371
26.4.1. Regular Expressions	371
26.4.2. Test Numbers	372
26.4.3. Labels	373
26.4.4. Repeating Tests	374
26.5. Parallel Execution	375
26.6. Managing Test Resources	377
26.6.1. Defining Test Resource Requirements	378
26.6.2. Specifying Available System Resources	379
26.6.3. Using Resources Allocated To A Test	381
26.7. Test Dependencies	382
26.8. Cross-compiling And Emulators	385

26.9. Build And Test Mode	386
26.10. CDash Integration	388
26.10.1. Key CDash Concepts	389
26.10.2. Executing Pipelines And Actions	390
26.10.3. CTest Configuration	392
26.10.4. Test Measurements And Results	396
26.11. Output Control	398
26.12. GoogleTest	400
26.13. Recommended Practices	405
27. Installing	408
27.1. Directory Layout	409
27.1.1. Relative Layout	409
27.1.2. Base Install Location	411
27.2. Installing Project Targets	413
27.2.1. Interface Properties	419
27.2.2. RPATH	420
27.2.3. Apple-specific Targets	423
27.3. Installing Exports	426
27.4. Installing Imported Targets	430
27.5. Installing Files	431
27.5.1. File Sets	431
27.5.2. Explicit Public And Private Headers	434
27.5.3. Simple Files And Programs	435
27.5.4. Whole Directories	436
27.6. Custom Install Logic	439
27.7. Installing Dependencies	440
27.7.1. Runtime Dependency Sets	440
27.7.2. InstallRequiredSystemLibraries Module	442
27.7.3. BundleUtilities And GetPrerequisites	443
27.8. Writing A Config Package File	443
27.8.1. Config Files For CMake Projects	444
27.8.2. Config Files For Non-CMake Projects	452
27.9. Executing An Install	454
27.10. Recommended Practices	454
28. Packaging	458
28.1. Packaging Basics	458
28.2. Components	463
28.3. Multi Configuration Packages	467
28.3.1. Multiple Build Directories	467
28.3.2. Pass Multiple Configurations To cpack	468
28.4. Package Generators	469

28.4.1. Simple Archives	470
28.4.2. Qt Installer Framework (IFW)	471
28.4.3. WIX	477
28.4.4. NSIS	479
28.4.5. DragNDrop	481
28.4.6. productbuild	483
28.4.7. RPM	485
28.4.8. DEB	489
28.4.9. FreeBSD	490
28.4.10. Cygwin	490
28.4.11. NuGet	491
28.4.12. External	491
28.5. Recommended Practices	491
29. ExternalProject	494
29.1. High Level Overview	494
29.2. Directory Layout	496
29.3. Built-in Steps	497
29.3.1. Archive Downloads	497
29.3.2. Repository Download Methods	499
29.3.3. Configuration Step	501
29.3.4. Build Step	502
29.3.5. Install Step	502
29.3.6. Test Step	503
29.4. Step Management	503
29.5. Miscellaneous Features	506
29.6. Common Issues	508
29.7. ExternalData	511
29.8. Recommended Practices	511
30. FetchContent	513
30.1. Comparison With ExternalProject	513
30.2. Basic Usage	514
30.3. Resolving Dependencies	515
30.4. Integration With find_package()	517
30.4.1. Try find_package() Before FetchContent	518
30.4.2. Redirect find_package() To FetchContent	520
30.4.3. Redirections Directory	521
30.5. Developer Overrides	524
30.6. Other Uses For FetchContent	526
30.7. Restrictions	528
30.8. Recommended Practices	528
31. Making Projects Consumable	530

31.1. Use Project-specific Names	530
31.2. Don't Assume A Top Level Build	532
31.3. Avoid Hard-coding Developer Choices	533
31.4. Avoid Package Variables If Possible	534
31.5. Use Appropriate Methods To Obtain Dependencies	535
31.6. Recommended Practices	536
32. Dependency Providers	537
32.1. Top Level Setup Injection Point	537
32.2. Dependency Provider Implementation	538
32.2.1. Accepting find_package() Requests	539
32.2.2. Accepting FetchContent_MakeAvailable() Requests	541
32.2.3. Accepting Multiple Request Methods	542
32.2.4. Wrapping The Built-in Implementations	543
32.2.5. Preserving Variable Values	543
32.2.6. Delegating Providers	544
32.3. Recommended Practices	545
33. Presets	547
33.1. High Level Structure	547
33.2. Configure Presets	549
33.2.1. Essential Fields	549
33.2.2. Inheritance	550
33.2.3. Macros	552
33.2.4. Environment Variables	553
33.2.5. Toolchains	553
33.2.6. Conditions	555
33.3. Build Presets	557
33.4. Test Presets	558
33.5. Package Presets	559
33.6. Workflow Presets	560
33.7. Recommended Practices	562
34. Project Organization	564
34.1. Superbuild Structure	564
34.2. Non-superbuild Structure	566
34.3. Common Top Level Subdirectories	568
34.4. IDE Projects	569
34.5. Defining Targets	572
34.5.1. Building Up A Target Across Directories	574
34.5.2. Target Output Locations	576
34.6. Windows-specific Issues	577
34.7. Cleaning Files	579
34.8. Re-running CMake On File Changes	579

34.9. Injecting Files Into Projects	580
34.10. Recommended Practices	581
35. Build Performance	584
35.1. Unity Builds	584
35.1.1. BATCH Mode	585
35.1.2. GROUP Mode	587
35.2. Precompiled Headers	588
35.3. Build Parallelism	589
35.3.1. Makefiles Generators	590
35.3.2. Ninja Generators	590
35.3.3. Visual Studio Generators	593
35.3.4. Xcode Generator	595
35.3.5. Optimizing Build Dependencies	595
35.4. Compiler Caches	597
35.4.1. Ccache Configuration	598
35.4.2. Makefiles And Ninja Generators	599
35.4.3. Xcode Generator	600
35.4.4. Visual Studio Generators	601
35.4.5. Combined Generator Support	602
35.5. Debug-related Improvements	602
35.6. Alternative Linkers	603
35.7. Recommended Practices	604
36. Working With Qt	607
36.1. Basic Setup	607
36.2. Build Details	609
36.2.1. Standard Library Implementation	609
36.2.2. Position Independent Code	609
36.2.3. Windows GUI Applications	610
36.3. Autogen	610
36.3.1. Moc	611
36.3.2. Widgets	615
36.3.3. Resources	616
36.4. Translations	618
36.5. Deployment	621
36.5.1. Qt Deployment Tools	621
36.5.2. Deploying Translation Files	623
36.5.3. Linux Considerations	625
36.6. Transition To Qt 6	625
36.7. Recommended Practices	626
Appendix A: Timer Dependency Provider	628
Appendix B: Full Compiler Cache Example	629

Chapter 4. Building Simple Targets

As shown in the previous chapter, it is relatively straightforward to define a simple executable in CMake. The simple example given previously required defining a target name for the executable and listing the source files to be compiled:

```
add_executable(MyApp main.cpp)
```

This assumes the developer wants a basic console executable to be built, but CMake also allows the developer to define other types of executables, such as app bundles on Apple platforms and Windows GUI applications. This chapter discusses additional options which can be given to `add_executable()` to specify these details.

In addition to executables, developers also frequently need to build and link libraries. CMake supports a few different kinds of libraries, including static, shared, modules and frameworks. CMake also offers very powerful features for managing dependencies between targets and how libraries are linked. This whole area of libraries and how to work with them in CMake forms the bulk of this chapter. The concepts covered here are used extensively throughout the remainder of this book. Some very basic use of variables and properties are also given to provide a flavor for how these CMake features relate to libraries and targets in general.

4.1. Executables

The more complete form of the basic `add_executable()` command is as follows:

```
add_executable(targetName [WIN32] [MACOSX_BUNDLE]
                    [EXCLUDE_FROM_ALL]
                    source1 [source2 ...]
)
```

The only differences to the form shown previously are the new optional keywords.

WIN32

When building the executable on a Windows platform, this option instructs CMake to build the executable as a Windows GUI application. In practice, this means it will be created with a `WinMain()` entry point instead of just `main()` and it will be linked with the `/SUBSYSTEM:WINDOWS` option. On all other platforms, the `WIN32` option is ignored.

MACOSX_BUNDLE

When present, this option directs CMake to build an app bundle when building on an Apple platform. Contrary to what the option name suggests, it applies not just to macOS, but also to other Apple platforms like iOS as well. The exact effects of this option vary somewhat between platforms. For example, on macOS, the app bundle layout has a very specific directory structure, whereas on iOS, the directory structure is flattened. CMake will also generate a basic `Info.plist` file for bundles. These and other details are covered in more detail in [Section 24.2, “Application Bundles”](#). On non-Apple platforms, the `MACOSX_BUNDLE` keyword is ignored.

EXCLUDE_FROM_ALL

Sometimes, a project defines a number of targets, but by default only some of them should be built. When no target is specified at build time, the default ALL target is built (depending on the CMake generator being used, the name may be slightly different, such as ALL_BUILD for Xcode). If an executable is defined with the EXCLUDE_FROM_ALL option, it will not be included in that default ALL target. The executable will then only be built if it is explicitly requested by the build command or if it is a dependency for another target that is part of the default ALL build. A common situation where it can be useful to exclude a target from ALL is where the executable is a developer tool that is only needed occasionally.

In addition to the above, there are other forms of the `add_executable()` command which produce a kind of reference to an existing executable or target rather than defining a new one to be built. These alias executables are covered in detail in [Chapter 18, Target Types](#).

4.2. Defining Libraries

Creating simple executables is a fundamental need of any build system. For many larger projects, the ability to create and work with libraries is also essential to keep the project manageable. CMake supports building a variety of different kinds of libraries, taking care of many of the platform differences, but still supporting the native idiosyncrasies of each. Library targets are defined using the `add_library()` command, of which there are a number of forms. The most basic of these is the following:

```
add_library(targetName [STATIC | SHARED | MODULE]
                  [EXCLUDE_FROM_ALL]
                  source1 [source2 ...]
)
```

This form is analogous to how `add_executable()` is used to define a simple executable. The `targetName` is used within the `CMakeLists.txt` file to refer to the library, with the name of the built library on the file system being derived from this name by default. The `EXCLUDE_FROM_ALL` keyword has exactly the same effect as it does for `add_executable()`, namely to prevent the library from being included in the default ALL target. The type of library to be built is specified by one of the remaining three keywords `STATIC`, `SHARED` or `MODULE`.

STATIC

Specifies a static library or archive. On Windows, the default library name would be `targetName.lib`, while on Unix-like platforms, it would typically be `libtargetName.a`.

SHARED

Specifies a shared or dynamically linked library. On Windows, the default library name would be `targetName.dll`, on Apple platforms it would be `libtargetName.dylib` and on other Unix-like platforms it would typically be `libtargetName.so`. On Apple platforms, shared libraries can also be marked as frameworks, a topic covered in [Section 24.3, “Frameworks”](#).

MODULE

Specifies a library that is somewhat like a shared library, but is intended to be loaded

dynamically at run-time rather than being linked directly to a library or executable. These are typically plugins or optional components the user may choose to be loaded or not. On Windows platforms, no import library is created for the DLL.

It is possible to omit the keyword defining what type of library to build. Unless the project specifically requires a particular type of library, the preferred practice is to not specify it and leave the choice up to the developer when building the project. In such cases, the library will be either `STATIC` or `SHARED`, with the choice determined by the value of a CMake variable called `BUILD_SHARED_LIBS`. If `BUILD_SHARED_LIBS` has been set to `true`, the library target will be a shared library, otherwise it will be static. Working with variables is covered in detail in [Chapter 5, Variables](#), but for now, one way to set this variable is by including a `-D` option on the `cmake` command line like so:

```
cmake -DBUILD_SHARED_LIBS=YES /path/to/source
```

It could be set in the `CMakeLists.txt` file instead with the following placed before any `add_library()` commands, but that would then require developers to modify it if they wanted to change it (i.e. it would be less flexible):

```
set(BUILD_SHARED_LIBS YES)
```

Just as for executables, library targets can also be defined to refer to some existing binary or target rather than being built by the project. Another type of pseudo-library is also supported for collecting together object files without going as far as creating a static library. These are all discussed in detail in [Chapter 18, Target Types](#).

4.3. Linking Targets

When considering the targets that make up a project, developers are typically used to thinking in terms of library A needing library B, so A is linked to B. This is the traditional way of looking at library handling, where the idea of one library needing another is very simplistic. In reality, however, there are a few different types of dependency relationships that can exist between libraries:

PRIVATE

Private dependencies specify that library A uses library B in its own internal implementation. Anything else that links to library A doesn't need to know about B because it is an internal implementation detail of A.

PUBLIC

Public dependencies specify that not only does library A use library B internally, it also uses B in its interface. This means that A cannot be used without B, so anything that uses A will also have a direct dependency on B. An example of this would be a function defined in library A which has at least one parameter of a type defined and implemented in library B, so code cannot call the function from A without providing a parameter whose type comes from B.

INTERFACE

Interface dependencies specify that in order to use library A, parts of library B must also be used. This differs from a public dependency in that library A doesn't require B internally, it only uses B in its interface. An example of where this is useful is when working with library targets defined using the `INTERFACE` form of `add_library()`, such as when using a target to represent a header-only library's dependencies (see [Section 18.2.4, “Interface Libraries”](#)).

CMake captures this richer set of dependency relationships with its `target_link_libraries()` command, not just the simplistic idea of needing to link. The general form of the command is:

```
target_link_libraries(targetName
    <PRIVATE|PUBLIC|INTERFACE> item1 [item2 ...]
    [<PRIVATE|PUBLIC|INTERFACE> item3 [item4 ...]]
    ...
)
```

This allows projects to precisely define how one library depends on others. CMake then takes care of managing the dependencies throughout the chain of libraries linked in this fashion. For example, consider the following:

```
add_library(Collector src1.cpp)
add_library(Algo src2.cpp)
add_library(Engine src3.cpp)
add_library(Ui src4.cpp)
add_executable(MyApp main.cpp)

target_link_libraries(Collector
    PUBLIC Ui
    PRIVATE Algo Engine
)
target_link_libraries(MyApp PRIVATE Collector)
```

In this example, the `Ui` library is linked to the `Collector` library as `PUBLIC`, so even though `MyApp` only directly links to `Collector`, `MyApp` will also be linked to `Ui` because of that `PUBLIC` relationship. The `Algo` and `Engine` libraries, on the other hand, are linked to `Collector` as `PRIVATE`, so `MyApp` will not be directly linked to them. [Section 18.2, “Libraries”](#) discusses additional behaviors for static libraries which may result in further linking to satisfy dependency relationships, including cyclic dependencies.

Later chapters present a few other `target_...()` commands which further enhance the dependency information carried between targets. These allow compiler/linker flags and header search paths to also carry through from one target to another when they are connected by `target_link_libraries()`. These features were added progressively from CMake 2.8.11 through to 3.2 and lead to considerably simpler and more robust `CMakeLists.txt` files.

Later chapters also discuss the use of more complex source directory hierarchies. In such cases, if using CMake 3.12 or earlier, the `targetName` used with `target_link_libraries()` must have been defined by an `add_executable()` or `add_library()` command in the same directory from which `target_link_libraries()` is being called (this restriction was removed in CMake 3.13).

4.4. Linking Non-targets

In the preceding section, all the items being linked to were existing CMake targets, but the `target_link_libraries()` command is more flexible than that. In addition to CMake targets, the following things can also be specified as items in a `target_link_libraries()` command:

Full path to a library file

CMake will add the library file to the linker command. If the library file changes, CMake will detect that change and re-link the target. Note that from CMake version 3.3, the linker command always uses the full path specified, but prior to version 3.3, there were some situations where CMake may ask the linker to search for the library instead (e.g. replace `/usr/lib/libfoo.so` with `-lfoo`). The reasoning and details of the pre-3.3 behavior are non-trivial and are largely historical, but for the interested reader, the full set of information is available in the CMake documentation under the `CMP0060` policy.

Plain library name

If just the name of the library is given with no path, the linker command will search for that library (e.g. `foo` becomes `-lfoo` or `foo.lib`, depending on the platform). This would be common for libraries provided by the system.

Link flag

As a special case, items starting with a hyphen other than `-l` or `-framework` will be treated as flags to be added to the linker command. The CMake documentation warns that these should only be used for `PRIVATE` items, since they would be carried through to other targets if defined as `PUBLIC` or `INTERFACE` and this may not always be safe.

4.5. Old-style CMake

For historical reasons, any link item specified in `target_link_libraries()` may be preceded by one of the keywords `debug`, `optimized` or `general`. The effect of these keywords is to further refine when the item following it should be included based on whether the build is configured as a debug build (see [Chapter 14, Build Type](#)). If an item is preceded by the `debug` keyword, then it will only be added if the build is a debug build. If an item is preceded by the `optimized` keyword, it will only be added if the build is not a debug build. The `general` keyword specifies that the item should be added for all build configurations, which is the default behavior anyway if no keyword is used. The `debug`, `optimized` and `general` keywords should be avoided for new projects as there are clearer, more flexible and more robust ways to achieve the same thing with today's CMake features.

The `target_link_libraries()` command also has a few other forms, some of which have been part of CMake from well before version 2.8.11. These forms are discussed here for the benefit of understanding older CMake projects, but their use is generally discouraged for new projects. The full form shown previously with `PRIVATE`, `PUBLIC` and `INTERFACE` sections should be preferred, as it expresses the nature of dependencies with more accuracy.

```
target_link_libraries(targetName item [item...])
```

The above form is generally equivalent to the items being defined as `PUBLIC`, but in certain situations, they may instead be treated as `PRIVATE`. In particular, if a project defines a chain of library dependencies with a mix of both old and new forms of the command, the old-style form will generally be treated as `PRIVATE`.

Another supported but deprecated form is the following:

```
target_link_libraries(targetName
    LINK_INTERFACE_LIBRARIES item [item...]
)
```

This is a pre-cursor to the `INTERFACE` keyword of the newer form covered above, but its use is discouraged by the CMake documentation. Its behavior can affect different target properties, with the policy settings controlling that behavior. This is a potential source of confusion for developers which can be avoided by using the newer `INTERFACE` form instead.

```
target_link_libraries(targetName
    <LINK_PRIVATE|LINK_PUBLIC> lib [lib...]
    [<LINK_PRIVATE|LINK_PUBLIC> lib [lib...]]
)
```

Similar to the previous old-style form, this one is a pre-cursor to the `PRIVATE` and `PUBLIC` keyword versions of the newer form. Again, the old-style form has the same confusion over which target properties it affects and the `PRIVATE/PUBLIC` keyword form should be preferred for new projects.

4.6. Recommended Practices

Target names need not be related to the project name. It is common to see tutorials and examples use a variable for the project name and reuse that variable for the name of an executable target:

```
# Poor practice, but very common
set(projectName MyExample)
project(${projectName})
add_executable(${projectName} ...)
```

This only works for the most basic of projects and encourages a number of bad habits. Consider the project name and executable name as being separate, even if initially they start out the same. Set the project name directly rather than via a variable, choose a target name according to what the target does rather than the project it is part of and assume the project will eventually need to define more than one target. This reinforces better habits which will be important when working on more complex multi-target projects.

When naming targets for libraries, resist the temptation to start or end the name with `lib`. On many platforms (i.e. just about all except Windows), a leading `lib` will be prefixed automatically when constructing the actual library name to make it conform to the platform's usual convention. If the target name already begins with `lib`, the library file names end up with the form `liblibsomething...`, which people often assume to be a mistake.

Unless there are strong reasons to do so, try to avoid specifying the `STATIC` or `SHARED` keyword for a library until it is known to be needed. This allows greater flexibility in choosing between static or dynamic libraries as an overall project-wide strategy. The `BUILD_SHARED_LIBS` variable can be used to change the default in one place instead of having to modify every call to `add_library()`.

Aim to always specify `PRIVATE`, `PUBLIC` and/or `INTERFACE` keywords when calling the `target_link_libraries()` command rather than following the old-style CMake syntax which assumed everything was `PUBLIC`. As a project grows in complexity, these three keywords have a stronger impact on how inter-target dependencies are handled. Using them from the beginning of a project also forces developers to think about the dependencies between targets, which can help to highlight structural problems within the project much earlier.

Chapter 14. Build Type

This chapter and the next cover two closely related topics. The build type (also known as the build configuration or build scheme in some IDE tools) is a high level control which selects different sets of compiler and linker behavior. Manipulation of the build type is the subject of this chapter, while the next chapter presents more specific details of controlling compiler and linker options. Together, these chapters cover material every CMake developer will typically use for all but the most trivial projects.

14.1. Build Type Basics

The build type has the potential to affect almost everything about the build in one way or another. While it primarily has a direct effect on the compiler and linker behavior, it also has an effect on the directory structure used for a project. This can in turn influence how a developer sets up their own local development environment, so the effects of the build type can be quite far-reaching.

Developers commonly think of builds as being one of two arrangements: debug or release. For a debug build, compiler flags are used to enable the recording of information that debuggers can use to associate machine instructions with the source code. Optimizations are frequently disabled in such builds so that the mapping from machine instruction to source code location is direct and easy to follow when stepping through program execution. A release build, on the other hand, generally has full optimizations enabled and no debug information generated.

These are examples of what CMake refers to as the *build type*. While projects are able to define whatever build types they want, the default build types provided by CMake are usually sufficient for most projects:

Debug

With no optimizations and full debug information, this is commonly used during development and debugging, as it typically gives the fastest build times and the best interactive debugging experience.

Release

This build type typically provides full optimizations for speed and no debug information, although some platforms may still generate debug symbols in certain circumstances. It is generally the build type used when building software for final production releases.

RelWithDebInfo

This is somewhat of a compromise of the previous two. It aims to give performance close to a Release build, but still allow some level of debugging. Most optimizations for speed are typically applied, but most debug functionality is also enabled. This build type is therefore most useful when the performance of a Debug build is not acceptable even for a debugging session. Note that the default settings for RelWithDebInfo will disable assertions.

MinSizeRel

This build type is typically only used for constrained resource environments such as embedded devices. The code is optimized for size rather than speed and no debug information is created.

Each build type results in a different set of compiler and linker flags. It may also change other behaviors, such as altering which source files get compiled or what libraries to link to. These details are covered in the next few sections, but before launching into those discussions, it is essential to understand how to select the build type and how to avoid some common problems.


14.1.1. Single Configuration Generators

Back in [Section 2.3, “Generating Project Files”](#), the different types of project generators were introduced. Some, like Makefiles and Ninja, support only a single build type per build directory. For these generators, the build type is chosen by setting the `CMAKE_BUILD_TYPE` cache variable. For example, to configure and then build a project with Ninja, one might use commands like this:

```
cmake -G Ninja -DCMAKE_BUILD_TYPE:STRING=Debug ../source
cmake --build .
```

The `CMAKE_BUILD_TYPE` cache variable can also be changed in the CMake GUI application instead of from the command line, but the end effect is the same. With CMake 3.22 or later, if the `CMAKE_BUILD_TYPE` cache variable is not set, it will be initialized from the `CMAKE_BUILD_TYPE` environment variable (if defined).

Rather than switching between different build types in the same build directory, an alternative strategy is to set up separate build directories for each build type, all still using the same sources. Such a directory structure might look something like this:



If frequently switching between build types, this arrangement avoids having to constantly recompile the same sources just because compiler flags change. It also allows a single configuration generator to effectively act like a multi configuration generator. IDE environments like Qt Creator support switching between build directories just as easily as Xcode or Visual Studio allow switching between build schemes or configurations.

14.1.2. Multiple Configuration Generators

Some generators, notably Xcode and Visual Studio, support multiple configurations in a single build directory. From CMake 3.17, the Ninja Multi-Config generator is also available. These multi-config generators ignore the `CMAKE_BUILD_TYPE` cache variable and instead require the developer to choose the build type within the IDE or with a command line option at build time. Configuring and building such projects would typically look something like this:

```
cmake -G Xcode ../source
cmake --build . --config Debug
```

When building within the Xcode IDE, the build type is controlled by the build scheme, while within the Visual Studio IDE, the current solution configuration controls the build type. Both environments keep separate directories for the different build types, so switching between builds doesn't cause constant rebuilds. In effect, the same thing is being done as the multiple build directory arrangement described above for single configuration generators, it's just that the IDE is handling the directory structure on the developer's behalf.

For command-line builds, the Ninja Multi-Config generator has a little more flexibility compared to the other multi-config generators. The `CMAKE_DEFAULT_BUILD_TYPE` cache variable can be used to change the default configuration to use when no configuration is specified on the build command line. The Xcode and Visual Studio generators have their own fixed logic for determining the default configuration in this scenario. The Ninja Multi-Config generator also supports advanced features that allow custom commands to execute as one configuration, but other targets to be built with one or more other configurations. Most projects would not typically need or benefit from these more advanced features, but the CMake documentation for the Ninja Multi-Config generator provides the essential details, with examples.

14.2. Common Errors

Note how for single configuration generators, the build type is specified at *configure* time, whereas for multi configuration generators, the build type is specified at *build* time. This distinction is critical, as it means the build type is not always known when CMake is processing a project's `CMakeLists.txt` file. Consider the following piece of CMake code, which unfortunately is rather common, but demonstrates an incorrect pattern:

```
# WARNING: Do not do this!
if(CMAKE_BUILD_TYPE STREQUAL "Debug")
    # Do something only for debug builds
endif()
```

The above would work fine for Makefile-based generators and Ninja, but not for Xcode, Visual Studio or Ninja Multi-Config. In practice, just about any logic based on `CMAKE_BUILD_TYPE` within a project is questionable unless it is protected by a check to confirm a single configuration generator is being used. For multi configuration generators, this variable is likely to be empty, but even if it isn't, its value should be considered unreliable because the build will ignore it. Rather than referring to `CMAKE_BUILD_TYPE` in the `CMakeLists.txt` file, projects should instead use other more robust alternative techniques, such as generator expressions based on `$<CONFIG:...>`.

When scripting builds, a common deficiency is to assume a particular generator is used or to not properly account for differences between single and multi configuration generators. Developers should ideally be able to change the generator in one place and the rest of the script should still function correctly. Conveniently, single configuration generators will ignore any build-time specification and multi configuration generators will ignore the `CMAKE_BUILD_TYPE` variable, so by specifying both, a script can account for both cases. For example:

```
mkdir build
cd build
cmake -G Ninja -DCMAKE_BUILD_TYPE=Release ../source
cmake --build . --config Release
```

With the above example, a developer could simply change the generator name given to the `-G` parameter and the rest of the script would work unchanged.

Not explicitly setting the `CMAKE_BUILD_TYPE` for single configuration generators is also common, but usually not what the developer intended. A behavior unique to single configuration generators is that if `CMAKE_BUILD_TYPE` is not set, the build type will usually be empty. This can lead to the misunderstanding that an empty build type is equivalent to Debug, but this is not so. An empty build type is its own unique, nameless build type. In such cases, no configuration-specific compiler or linker flags are used, which often results in invoking the compiler and linker with minimal flags. The behavior is then determined by the compiler's and linker's own defaults. While this may often be similar to the Debug build type's behavior, it is by no means guaranteed.

Using the Visual Studio compilers with a single configuration generator is somewhat of a special case. For that toolchain, there are different runtime libraries for debug and non-debug builds. An empty build type would make it unclear which runtime should be used. To avoid this ambiguity, the build type will default to Debug for this combination.

14.3. Custom Build Types

Sometimes a project may want to limit the set of build types to a subset of the defaults, or it may want to add other custom build types with a special set of compiler and linker flags. A good example of the latter is adding a build type for profiling or code coverage, both of which require specific compiler and linker settings.

There are two main places where a developer may see the set of build types. When using IDE environments for multi configuration generators like Xcode and Visual Studio, the IDE provides a drop-down list or similar from which the developer selects the configuration they wish to build. For single configuration generators like Makefiles or Ninja, the build type is entered directly for the `CMAKE_BUILD_TYPE` cache variable, but the CMake GUI application can be made to present a combo box of valid choices instead of a simple text edit field. The mechanisms behind these two cases are different, so they must be handled separately.

The set of build types known to multi configuration generators is controlled by the `CMAKE_CONFIGURATION_TYPES` cache variable, or more accurately, by the value of this variable at the end of processing the top level `CMakeLists.txt` file. The first encountered `project()` command populates the cache variable if it has not already been defined. With CMake 3.22 or later, a `CMAKE_CONFIGURATION_TYPES` environment variable can provide the default value. If that environment variable isn't set or an earlier CMake version is used, the default value will be (possibly a subset of) the four standard configurations mentioned in [Section 14.1, “Build Type Basics”](#) (Debug, Release, RelWithDebInfo and MinSizeRel).

Projects may modify the `CMAKE_CONFIGURATION_TYPES` variable after the first `project()` command, but only in the top level `CMakeLists.txt` file. Some CMake generators rely on this variable having a

consistent value throughout the whole project. Custom build types can be defined by adding them to `CMAKE_CONFIGURATION_TYPES` and unwanted build types can be removed from that list. Note that only the non-cache variable should be modified, as changing the cache variable may discard changes made by the developer.

Care needs to be taken to avoid setting `CMAKE_CONFIGURATION_TYPES` if it is not already defined. Prior to CMake 3.9, a very common approach for determining whether a multi configuration generator was being used was to check if `CMAKE_CONFIGURATION_TYPES` was non-empty. Even parts of CMake itself did this prior to 3.11. While this method is usually accurate, it is not unusual to see projects unilaterally set `CMAKE_CONFIGURATION_TYPES` even if using a single configuration generator. This can lead to wrong decisions being made regarding the type of generator in use. To address this, CMake 3.9 added a new `GENERATOR_IS_MULTI_CONFIG` global property which is set to true when a multi configuration generator is being used, providing a definitive way to obtain that information instead of relying on inferring it from `CMAKE_CONFIGURATION_TYPES`. Even so, checking `CMAKE_CONFIGURATION_TYPES` is still such a prevalent pattern that projects should continue to only modify it if it exists and never create it themselves. It should also be noted that prior to CMake 3.11, adding custom build types to `CMAKE_CONFIGURATION_TYPES` was not safe. Certain parts of CMake only accounted for the default build types, but even so, projects may still be able to usefully define custom build types with earlier CMake versions, depending on how they are going to be used. That said, for better robustness, it is recommended that at least CMake 3.11 be used if custom build types are going to be defined.

Another aspect of this issue is that developers may add their own types to the `CMAKE_CONFIGURATION_TYPES` cache variable and/or remove those they are not interested in. Projects should therefore not make any assumptions about what configuration types are or are not defined.

Taking the above points into account, the following pattern shows the preferred way for projects to add their own custom build types for multi configuration generators:

```
cmake_minimum_required(3.11)
project(Foo)

# Only make changes if we are the top level project
if(CMAKE_SOURCE_DIR STREQUAL CMAKE_CURRENT_SOURCE_DIR)
    get_property(isMultiConfig GLOBAL
        PROPERTY GENERATOR_IS_MULTI_CONFIG
    )
    if(isMultiConfig)
        if(NOT "Profile" IN_LIST CMAKE_CONFIGURATION_TYPES)
            list(APPEND CMAKE_CONFIGURATION_TYPES Profile)
        endif()
    endif()
    # Set Profile-specific flag variables as needed...
endif()
```

For single configuration generators, there is only one build type. This is specified by the `CMAKE_BUILD_TYPE` cache variable, which is a string. In the CMake GUI, this is normally presented as a text edit field, so the developer can edit it to contain whatever arbitrary content they wish. As discussed back in [Section 9.6, “Cache Variable Properties”](#), cache variables can have their `STRINGS` property defined to hold a set of valid values. The CMake GUI application will then present that variable as a combo box containing the valid values instead of as a text edit field.

```

set_property(CACHE CMAKE_BUILD_TYPE PROPERTY
    STRINGS Debug Release Profile
)

```

Properties can only be changed from within the project's CMakeLists.txt files, so they can safely set the STRINGS property without having to worry about preserving any developer changes. Note, however, that setting the STRINGS property of a cache variable does not guarantee that the cache variable will hold one of the defined values, it only controls how the variable is presented in the CMake GUI application. Developers can still set CMAKE_BUILD_TYPE to any value at the cmake command line or edit the CMakeCache.txt file manually. In order to rigorously require the variable to have one of the defined values, a project must explicitly perform that test itself.

```

set(allowedBuildTypes Debug Release Profile)

# WARNING: This logic is not sufficient
if(NOT CMAKE_BUILD_TYPE IN_LIST allowedBuildTypes)
    message(FATAL_ERROR "${CMAKE_BUILD_TYPE} is not a known build type")
endif()

```

The default value for CMAKE_BUILD_TYPE is an empty string, so the above would cause a fatal error for both single and multi configuration generators unless the developer explicitly set it. This is undesirable, especially for multi configuration generators which don't even use the CMAKE_BUILD_TYPE variable's value. This can be handled by having the project provide a default value if CMAKE_BUILD_TYPE hasn't been set. Furthermore, the techniques for multi and single configuration generators can and should be combined to give robust behavior across all generator types. The end result would look something like this:

```

cmake_minimum_required(3.11)
project(Foo)

if(CMAKE_SOURCE_DIR STREQUAL CMAKE_CURRENT_SOURCE_DIR)
    get_property(isMultiConfig GLOBAL PROPERTY GENERATOR_IS_MULTI_CONFIG)
    if(isMultiConfig)
        if(NOT "Profile" IN_LIST CMAKE_CONFIGURATION_TYPES)
            list(APPEND CMAKE_CONFIGURATION_TYPES Profile)
        endif()
    else()
        set(allowedBuildTypes Debug Release Profile)
        set_property(CACHE CMAKE_BUILD_TYPE PROPERTY
            STRINGS "${allowedBuildTypes}")
    )
    if(NOT CMAKE_BUILD_TYPE)
        set(CMAKE_BUILD_TYPE Debug CACHE STRING "" FORCE)
    elseif(NOT CMAKE_BUILD_TYPE IN_LIST allowedBuildTypes)
        message(FATAL_ERROR "Unknown build type: ${CMAKE_BUILD_TYPE}")
    endif()
endif()

# Set relevant Profile-specific flag variables as needed...
endif()

```

The above techniques enable *selecting* a custom build type, but they don't *define* anything about that build type. Selecting a build type specifies which configuration-specific variables to use. It also affects any generator expressions whose logic depends on the current configuration (`<CONFIG>` and `<CONFIG:··>`). These variables and generator expressions are discussed in the next chapter. For now, the following two families of variables are of primary interest.

- `CMAKE_<LANG>_FLAGS_<CONFIG>`
- `CMAKE_<TARGETTYPE>_LINKER_FLAGS_<CONFIG>`

The flags specified in these variables are added to the default set provided by the same-named variables without the `_<CONFIG>` suffix. A custom Profile build type might be defined like so:

```
set(CMAKE_C_FLAGS_PROFILE          "-p -g -O2" CACHE STRING "")
set(CMAKE_CXX_FLAGS_PROFILE        "-p -g -O2" CACHE STRING "")
set(CMAKE_EXE_LINKER_FLAGS_PROFILE "-p -g -O2" CACHE STRING "")
set(CMAKE_SHARED_LINKER_FLAGS_PROFILE "-p -g -O2" CACHE STRING "")
set(CMAKE_STATIC_LINKER_FLAGS_PROFILE "" CACHE STRING "")
set(CMAKE_MODULE_LINKER_FLAGS_PROFILE "-p -g -O2" CACHE STRING "")
```

The above assumes a GCC-compatible compiler to keep the example simple and turns on profiling as well as enabling debugging symbols and most optimizations. An alternative is to base the compiler and linker flags on one of the other build types and add the extra flags needed. This can be done as long as it comes after the `project()` command, since that command populates the default compiler and linker flag variables. For profiling, the `RelWithDebInfo` default build type is a good one to choose as the base configuration since it enables both debugging and most optimizations:

```
set(CMAKE_C_FLAGS_PROFILE
    "${CMAKE_C_FLAGS_RELWITHDEBINFO} -p"
    CACHE STRING ""
)
set(CMAKE_CXX_FLAGS_PROFILE
    "${CMAKE_CXX_FLAGS_RELWITHDEBINFO} -p"
    CACHE STRING ""
)
set(CMAKE_EXE_LINKER_FLAGS_PROFILE
    "${CMAKE_EXE_LINKER_FLAGS_RELWITHDEBINFO} -p"
    CACHE STRING ""
)
set(CMAKE_SHARED_LINKER_FLAGS_PROFILE
    "${CMAKE_SHARED_LINKER_FLAGS_RELWITHDEBINFO} -p"
    CACHE STRING ""
)
set(CMAKE_STATIC_LINKER_FLAGS_PROFILE
    "${CMAKE_STATIC_LINKER_FLAGS_RELWITHDEBINFO}"
    CACHE STRING ""
)
set(CMAKE_MODULE_LINKER_FLAGS_PROFILE
    "${CMAKE_MODULE_LINKER_FLAGS_RELWITHDEBINFO} -p"
    CACHE STRING ""
)
```


Each custom configuration should have the associated compiler and linker flag variables defined. For some multi configuration generator types, CMake will check that the required variables exist and will fail with an error if they are not set.

Another variable which may sometimes be defined for a custom build type is `CMAKE_<CONFIG>_POSTFIX`. It is used to initialize the `<CONFIG>_POSTFIX` property of each library target, with its value being appended to the file name of such targets when built for the specified configuration. This allows libraries from multiple build types to be put in the same directory without overwriting each other. `CMAKE_DEBUG_POSTFIX` is often set to values like `d` or `_debug`, especially for Visual Studio builds where different runtime DLLs must be used for Debug and non-Debug builds, so packages may need to include libraries for both build types. In the case of the custom Profile build type defined above, an example might be:

```
set(CMAKE_PROFILE_POSTFIX _profile)
```

If creating packages that contain multiple build types, setting `CMAKE_<CONFIG>_POSTFIX` for each build type is highly recommended. By convention, the postfix for Release builds is typically empty. Note though that the `<CONFIG>_POSTFIX` target property is ignored on Apple platforms.

For historical reasons, the items passed to the `target_link_libraries()` command can be prefixed with the `debug` or `optimized` keywords to indicate that the named item should only be linked in for debug or non-debug builds respectively. A build type is considered to be a debug build if it is listed in the `DEBUG_CONFIGURATIONS` global property, otherwise it is considered to be optimized. For custom build types, they should have their name added to this global property if they should be treated as a debug build in this scenario. As an example, if a project defines its own custom build type called `StrictChecker` and that build type should be considered a non-optimized debug build type, it can (and should) make this clear like so:

```
set_property(GLOBAL APPEND PROPERTY
  DEBUG_CONFIGURATIONS StrictChecker
)
```

New projects should normally prefer to use generator expressions instead of the `debug` and `optimized` keywords with the `target_link_libraries()` command. The next chapter discusses this area in more detail.

14.4. Recommended Practices

Developers should not assume a particular CMake generator is being used to build their project. Another developer on the same project may prefer to use a different generator because it integrates better with their IDE tool, or a future version of CMake may add support for a new generator type which might bring other benefits. Certain build tools may contain bugs which a project may later be affected by, so it can be useful to have alternative generators to fall back on until such bugs are fixed. Expanding a project's set of supported platforms can also be hindered if a particular CMake generator has been assumed.

When using single configuration generators like Makefiles or Ninja, consider using multiple build directories, one for each build type of interest. This allows switching between build types without forcing a complete recompile each time. This provides similar behavior to that inherently offered by multi configuration generators and can be a useful way to enable IDE tools like Qt Creator to simulate multi configuration functionality.

For single configuration generators, consider setting `CMAKE_BUILD_TYPE` to a better default value if it is empty. While an empty build type is technically valid, it is also often misunderstood by developers to mean a Debug build rather than its own distinct build type. Furthermore, avoid creating logic based on `CMAKE_BUILD_TYPE` unless it is first confirmed that a single configuration generator is being used. Even then, such logic is likely to be fragile and could probably be expressed with more generality and robustness using generator expressions instead.

Only consider modifying the `CMAKE_CONFIGURATION_TYPES` variable if it is known that a multi configuration generator is being used or if the variable already exists. If adding a custom build type or removing one of the default build types, do not modify the cache variable but instead change the regular variable of the same name (it will take precedence over the cache variable). Also prefer to add and remove individual items rather than completely replacing the list. Both of these measures will help avoid interfering with changes made to the cache variable by the developer. Only make such changes in the top level `CMakeLists.txt` file.

If requiring CMake 3.9 or later, use the `GENERATOR_IS_MULTI_CONFIG` global property to definitively query the generator type instead of relying on the existence of `CMAKE_CONFIGURATION_TYPES` to perform a less robust check.

A common but incorrect practice is to query the `LOCATION` target property to work out a target's output file name. A related error is to assume a particular build output directory structure in custom commands (see [Chapter 19, Custom Tasks](#)). These methods do not work for all build types, since `LOCATION` is not known at configure time for multi configuration generators and the build output directory structure is typically different across the various CMake generator types. Generator expressions like `$<TARGET_FILE:…>` should be used instead, as they robustly provide the required path for all generators, whether they be single or multi configuration.